#include <Servo.h>

#define KANALEN              1  // Aantal kanalen is momenteel 1.

#define MONITOR              0  // 0 = UIT; 1 = AAN.

#define KANAALPIN            4

#define SYNCTIME          4000  // Kanaal synchronisatie tijd in microseconden.

#define LEDROODPIN           6

#define LEDGEELPIN           9

#define DEURSERVOPINL        8

#define DEURSERVOPINR       10

#define KLEPSERVOPIN        12

#define OPENEN               1

#define SLUITEN              0

#define DEURSERVOHOEKMINL   40  // Hoek bij gesloten deur links

#define DEURSERVOHOEKMAXL  123  // Hoek bij geopende deur links.

#define DEURSERVOHOEKMIDL   60  // Hoek van de linkerdeur, waarbij de klep dicht gaat.

#define DEURSERVOHOEKMAXR  150  // Hoek bij gesloten deur rechts.

#define DEURSERVOHOEKMINR   60  // Hoek bij geopende deur rechts.

#define DEURSERVOHOEKMIDR   60  // Hoek van de rechterdeur, waarbij de klep dicht gaat.

#define KLEPSERVOHOEKMAX   150  // Hoek bij gesloten klep.

#define KLEPSERVOHOEKMIN    20  // Hoek bij geopende klep.

#define KLEPSERVOHOEKMID    60  // Hoek van de klep, waarbij de deuren open gaan

#define SERVOUPDATE         10  // Deze regel definieert de tijd in milliseconden tussen de servo stappen.

#define LEDUPDATE          400  // Deze regel definieert de knippertijd van de LED's in milliseconden.

int           Kanaal[KANALEN];                // Pulse waarden.

int           Waarde[KANALEN];                // Geconverteerde pulse waarden.

int           LedState             = LOW;     // Setting LED.

byte          Status               = SLUITEN; // Initiele toestand.

unsigned long ServoTimer           = millis () + SERVOUPDATE;

unsigned long LEDTimer             = millis () + LEDUPDATE;

byte          DeurServoHoekL       = DEURSERVOHOEKMINL;

byte          DeurServoHoekR       = DEURSERVOHOEKMAXR;

byte          KlepServoHoek        = KLEPSERVOHOEKMAX;

byte          DeurServoHoekTargetL = DEURSERVOHOEKMINL;

byte          DeurServoHoekTargetR = DEURSERVOHOEKMAXR;

byte          KlepServoHoekTarget  = KLEPSERVOHOEKMAX;

Servo         DeurServoL;

Servo         DeurServoR;

Servo         KlepServo;

/************************************************************************************************************/

/* Setup (eenmalig)                                                                                         */

/************************************************************************************************************/

void setup() 

  {

  Serial.begin (9600);                     // Start seriele monitor.

  pinMode (KANAALPIN, INPUT);              // Sluit het kanaal aan op een input pin.

  pinMode (LEDGEELPIN, OUTPUT);            // Sluit de gele LED aan op een output pin.

  pinMode (LEDROODPIN, OUTPUT);            // Sluit de rode LED aan op een output pin.

  digitalWrite (LEDGEELPIN, LOW);          // Zet de gele LED uit.

  digitalWrite (LEDROODPIN, LOW);          // Zet de rode LED uit.

  DeurServoL.attach (DEURSERVOPINL);       // Sluit linkerdeur servo aan op een output pin.

  DeurServoR.attach (DEURSERVOPINR);       // Sluit rechterdeur servo aan op een output pin.

  KlepServo.attach (KLEPSERVOPIN);         // Sluit klepservo aan op een output pin.

  DeurServoL.write (DeurServoHoekTargetL); // Sluit de linkerdeur.

  DeurServoR.write (DeurServoHoekTargetR); // Sluit de rechterdeur.

  KlepServo.write (KlepServoHoekTarget);   // Sluit de klep.

  }

/************************************************************************************************************/

/* Main loop (continu)                                                                                      */

/************************************************************************************************************/

void loop()

  {

  if (pulseIn (KANAALPIN, HIGH) > SYNCTIME);            // Wacht tot synchronisatie plaats heeft gevonden > 4 ms.

    {

    for (int i = 0; i < KANALEN; i++)                   // Leest de pulsen van de volgende kanalen.

      Kanaal[i] = pulseIn (KANAALPIN, HIGH);

    for (int i = 0; i < KANALEN; i++)

      {

      Waarde[i] = map (Kanaal[i], 1000, 2000, 0, 1000); // Converteer pulse waarde naar range van 0 tot 1000.

      if (MONITOR)                                      // Print alle gelezen waarden.

        {

        Serial.print ("CH ");

        Serial.print (i + 1);            // Print kanaalnummer.

        Serial.print (": ");

        Serial.print (Kanaal[i]);        // Print de pulse waarde.

        Serial.print (" Waarde : ");

        Serial.println (Waarde[i]);      // Print de geconverteerde waarde. 

        }

      }

    if (millis () > LEDTimer)

      {

      LEDTimer = millis () + LEDUPDATE;

      LedState = (LedState == LOW) ? HIGH : LOW;  // als de LED uit is, aanzetten en v.v.

      if (Waarde[0] < 400)                        // Linkerdeur sluiten, als deur half dicht is dan klep .

        {

        digitalWrite(LEDGEELPIN, LedState);

        DeurServoHoekTargetL = DEURSERVOHOEKMINL;

        DeurServoHoekTargetR = DEURSERVOHOEKMAXR;

        KlepServoHoekTarget  = KLEPSERVOHOEKMAX;

        Status               = SLUITEN;

        }

      else if (Waarde[0] > 600)                   // Deuren openen als klep half open is.

        {

        digitalWrite(LEDROODPIN, LedState);

        DeurServoHoekTargetL = DEURSERVOHOEKMAXL;

        DeurServoHoekTargetR = DEURSERVOHOEKMINR;     

        KlepServoHoekTarget  = KLEPSERVOHOEKMIN;

        Status               = OPENEN;

        }

      else

        {

        digitalWrite(LEDGEELPIN, LOW);

        digitalWrite(LEDROODPIN, LOW);

        }

      }

    }

  if (millis () > ServoTimer)

    {

    ServoTimer = millis () + SERVOUPDATE;

    switch (Status)

      {

      case OPENEN:  if (KlepServoHoek > KlepServoHoekTarget)

                      KlepServoHoek--;  

                    if ((DeurServoHoekR > DeurServoHoekTargetR) && (KlepServoHoek <= KLEPSERVOHOEKMID))

                      DeurServoHoekR--;

                    if ((DeurServoHoekL < DeurServoHoekTargetL) && (KlepServoHoek <= KLEPSERVOHOEKMID))

                      DeurServoHoekL++;

                    break;

      case SLUITEN: if (DeurServoHoekL > DeurServoHoekTargetL)

                      DeurServoHoekL--;

                    if (DeurServoHoekR < DeurServoHoekTargetR)

                      DeurServoHoekR++;

                    if ((KlepServoHoek < KlepServoHoekTarget) &&

                        (DeurServoHoekL <= DEURSERVOHOEKMIDL) && (DeurServoHoekR >= DEURSERVOHOEKMIDR))

                      KlepServoHoek++;

                    break;

      }

    DeurServoL.write (DeurServoHoekL);

    DeurServoR.write (DeurServoHoekR);

    KlepServo.write (KlepServoHoek);

    }

  }

/************************************************************************************************************/

/* Main loop (einde)                                                                                        */

/************************************************************************************************************/
